// File: Address.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.6;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize, which returns 0 for contracts in
        // construction, since the code is only stored at the end of the
        // constructor execution.

        uint256 size;
        assembly {
            size := extcodesize(account)
        }
        return size > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}
// File: Base64.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.6;

/// @title Base64
/// @notice Provides a function for encoding some bytes in base64
/// @author Brecht Devos <brecht@loopring.org>
library Base64 {
    bytes internal constant TABLE = "ABCDEFGHIJKLMNOPQRSTUVWXYZabcdefghijklmnopqrstuvwxyz0123456789+/";

    /// @notice Encodes some bytes to the base64 representation
    function encode(bytes memory data) internal pure returns (string memory) {
        uint256 len = data.length;
        if (len == 0) return "";

        // multiply by 4/3 rounded up
        uint256 encodedLen = 4 * ((len + 2) / 3);

        // Add some extra buffer at the end
        bytes memory result = new bytes(encodedLen + 32);

        bytes memory table = TABLE;

        assembly {
            let tablePtr := add(table, 1)
            let resultPtr := add(result, 32)

            for {
                let i := 0
            } lt(i, len) {

            } {
                i := add(i, 3)
                let input := and(mload(add(data, i)), 0xffffff)

                let out := mload(add(tablePtr, and(shr(18, input), 0x3F)))
                out := shl(8, out)
                out := add(out, and(mload(add(tablePtr, and(shr(12, input), 0x3F))), 0xFF))
                out := shl(8, out)
                out := add(out, and(mload(add(tablePtr, and(shr(6, input), 0x3F))), 0xFF))
                out := shl(8, out)
                out := add(out, and(mload(add(tablePtr, and(input, 0x3F))), 0xFF))
                out := shl(224, out)

                mstore(resultPtr, out)

                resultPtr := add(resultPtr, 4)
            }

            switch mod(len, 3)
            case 1 {
                mstore(sub(resultPtr, 2), shl(240, 0x3d3d))
            }
            case 2 {
                mstore(sub(resultPtr, 1), shl(248, 0x3d))
            }

            mstore(result, encodedLen)
        }

        return string(result);
    }
}
// File: Context.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.6;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}
// File: ERC165.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.6;

import "./IERC165.sol";

/**
 * @dev Implementation of the {IERC165} interface.
 *
 * Contracts that want to implement ERC165 should inherit from this contract and override {supportsInterface} to check
 * for the additional interface id that will be supported. For example:
 *
 * ```solidity
 * function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
 *     return interfaceId == type(MyInterface).interfaceId || super.supportsInterface(interfaceId);
 * }
 * ```
 *
 * Alternatively, {ERC165Storage} provides an easier to use but more expensive implementation.
 */
abstract contract ERC165 is IERC165 {
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IERC165).interfaceId;
    }
}
// File: ERC721.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.6;

import "./IERC721.sol";
import "./IERC721Receiver.sol";
import "./IERC721Metadata.sol";
import "./Address.sol";
import "./Context.sol";
import "./Strings.sol";
import "./ERC165.sol";

/**
 * @dev Implementation of https://eips.ethereum.org/EIPS/eip-721[ERC721] Non-Fungible Token Standard, including
 * the Metadata extension, but not including the Enumerable extension, which is available separately as
 * {ERC721Enumerable}.
 */
contract ERC721 is Context, ERC165, IERC721, IERC721Metadata {
    using Address for address;
    using Strings for uint256;

    // Token name
    string private _name;

    // Token symbol
    string private _symbol;

    // Mapping from token ID to owner address
    mapping(uint256 => address) internal _owners;

    // Mapping owner address to token count
    mapping(address => uint256) internal _balances;

    // Mapping from token ID to approved address
    mapping(uint256 => address) internal _tokenApprovals;

    // Mapping from owner to operator approvals
    mapping(address => mapping(address => bool)) private _operatorApprovals;

    /**
     * @dev Initializes the contract by setting a `name` and a `symbol` to the token collection.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override(ERC165, IERC165) returns (bool) {
        return
            interfaceId == type(IERC721).interfaceId ||
            interfaceId == type(IERC721Metadata).interfaceId ||
            super.supportsInterface(interfaceId);
    }

    /**
     * @dev See {IERC721-balanceOf}.
     */
    function balanceOf(address owner) public view virtual override returns (uint256) {
        require(owner != address(0), "ERC721: balance query for the zero address");
        return _balances[owner];
    }

    /**
     * @dev See {IERC721-ownerOf}.
     */
    function ownerOf(uint256 tokenId) public view virtual override returns (address) {
        address owner = _owners[tokenId];
        require(owner != address(0), "ERC721: owner query for nonexistent token");
        return owner;
    }

    /**
     * @dev See {IERC721Metadata-name}.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev See {IERC721Metadata-symbol}.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev See {IERC721Metadata-tokenURI}.
     */
    function tokenURI(uint256 tokenId) public view virtual override returns (string memory) {
        require(_exists(tokenId), "ERC721Metadata: URI query for nonexistent token");

        string memory baseURI = _baseURI();
        return bytes(baseURI).length > 0 ? string(abi.encodePacked(baseURI, tokenId.toString())) : "";
    }

    /**
     * @dev Base URI for computing {tokenURI}. If set, the resulting URI for each
     * token will be the concatenation of the `baseURI` and the `tokenId`. Empty
     * by default, can be overriden in child contracts.
     */
    function _baseURI() internal view virtual returns (string memory) {
        return "";
    }

    /**
     * @dev See {IERC721-approve}.
     */
    function approve(address to, uint256 tokenId) public virtual override {
        address owner = ERC721.ownerOf(tokenId);
        require(to != owner, "ERC721: approval to current owner");

        require(
            _msgSender() == owner || isApprovedForAll(owner, _msgSender()),
            "ERC721: approve caller is not owner nor approved for all"
        );

        _approve(to, tokenId);
    }

    /**
     * @dev See {IERC721-getApproved}.
     */
    function getApproved(uint256 tokenId) public view virtual override returns (address) {
        require(_exists(tokenId), "ERC721: approved query for nonexistent token");

        return _tokenApprovals[tokenId];
    }

    /**
     * @dev See {IERC721-setApprovalForAll}.
     */
    function setApprovalForAll(address operator, bool approved) public virtual override {
        require(operator != _msgSender(), "ERC721: approve to caller");

        _operatorApprovals[_msgSender()][operator] = approved;
        emit ApprovalForAll(_msgSender(), operator, approved);
    }

    /**
     * @dev See {IERC721-isApprovedForAll}.
     */
    function isApprovedForAll(address owner, address operator) public view virtual override returns (bool) {
        return _operatorApprovals[owner][operator];
    }

    /**
     * @dev See {IERC721-transferFrom}.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        //solhint-disable-next-line max-line-length
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: transfer caller is not owner nor approved");

        _transfer(from, to, tokenId);
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        safeTransferFrom(from, to, tokenId, "");
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) public virtual override {
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: transfer caller is not owner nor approved");
        _safeTransfer(from, to, tokenId, _data);
    }

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * `_data` is additional data, it has no specified format and it is sent in call to `to`.
     *
     * This internal function is equivalent to {safeTransferFrom}, and can be used to e.g.
     * implement alternative mechanisms to perform token transfer, such as signature-based.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeTransfer(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) internal virtual {
        _transfer(from, to, tokenId);
        require(_checkOnERC721Received(from, to, tokenId, _data), "ERC721: transfer to non ERC721Receiver implementer");
    }

    /**
     * @dev Returns whether `tokenId` exists.
     *
     * Tokens can be managed by their owner or approved accounts via {approve} or {setApprovalForAll}.
     *
     * Tokens start existing when they are minted (`_mint`),
     * and stop existing when they are burned (`_burn`).
     */
    function _exists(uint256 tokenId) internal view virtual returns (bool) {
        return _owners[tokenId] != address(0);
    }

    /**
     * @dev Returns whether `spender` is allowed to manage `tokenId`.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function _isApprovedOrOwner(address spender, uint256 tokenId) internal view virtual returns (bool) {
        require(_exists(tokenId), "ERC721: operator query for nonexistent token");
        address owner = ERC721.ownerOf(tokenId);
        return (spender == owner || getApproved(tokenId) == spender || isApprovedForAll(owner, spender));
    }

    /**
     * @dev Safely mints `tokenId` and transfers it to `to`.
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeMint(address to, uint256 tokenId) internal virtual {
        _safeMint(to, tokenId, "");
    }

    /**
     * @dev Same as {xref-ERC721-_safeMint-address-uint256-}[`_safeMint`], with an additional `data` parameter which is
     * forwarded in {IERC721Receiver-onERC721Received} to contract recipients.
     */
    function _safeMint(
        address to,
        uint256 tokenId,
        bytes memory _data
    ) internal virtual {
        _mint(to, tokenId);
        require(
            _checkOnERC721Received(address(0), to, tokenId, _data),
            "ERC721: transfer to non ERC721Receiver implementer"
        );
    }

    /**
     * @dev Mints `tokenId` and transfers it to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {_safeMint} whenever possible
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - `to` cannot be the zero address.
     *
     * Emits a {Transfer} event.
     */
    function _mint(address to, uint256 tokenId) internal virtual {
        require(to != address(0), "ERC721: mint to the zero address");
        require(!_exists(tokenId), "ERC721: token already minted");

        _beforeTokenTransfer(address(0), to, tokenId);

        _balances[to] += 1;
        _owners[tokenId] = to;

        emit Transfer(address(0), to, tokenId);
    }

    /**
     * @dev Destroys `tokenId`.
     * The approval is cleared when the token is burned.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     *
     * Emits a {Transfer} event.
     */
    function _burn(uint256 tokenId) internal virtual {
        address owner = ERC721.ownerOf(tokenId);

        _beforeTokenTransfer(owner, address(0), tokenId);

        // Clear approvals
        _approve(address(0), tokenId);

        _balances[owner] -= 1;
        delete _owners[tokenId];

        emit Transfer(owner, address(0), tokenId);
    }

    /**
     * @dev Transfers `tokenId` from `from` to `to`.
     *  As opposed to {transferFrom}, this imposes no restrictions on msg.sender.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     *
     * Emits a {Transfer} event.
     */
    function _transfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {
        require(ERC721.ownerOf(tokenId) == from, "ERC721: transfer of token that is not own");
        require(to != address(0), "ERC721: transfer to the zero address");

        _beforeTokenTransfer(from, to, tokenId);

        // Clear approvals from the previous owner
        _approve(address(0), tokenId);

        _balances[from] -= 1;
        _balances[to] += 1;
        _owners[tokenId] = to;

        emit Transfer(from, to, tokenId);
    }

    /**
     * @dev Approve `to` to operate on `tokenId`
     *
     * Emits a {Approval} event.
     */
    function _approve(address to, uint256 tokenId) internal virtual {
        _tokenApprovals[tokenId] = to;
        emit Approval(ERC721.ownerOf(tokenId), to, tokenId);
    }

    /**
     * @dev Internal function to invoke {IERC721Receiver-onERC721Received} on a target address.
     * The call is not executed if the target address is not a contract.
     *
     * @param from address representing the previous owner of the given token ID
     * @param to target address that will receive the tokens
     * @param tokenId uint256 ID of the token to be transferred
     * @param _data bytes optional data to send along with the call
     * @return bool whether the call correctly returned the expected magic value
     */
    function _checkOnERC721Received(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) private returns (bool) {
        if (to.isContract()) {
            try IERC721Receiver(to).onERC721Received(_msgSender(), from, tokenId, _data) returns (bytes4 retval) {
                return retval == IERC721Receiver.onERC721Received.selector;
            } catch (bytes memory reason) {
                if (reason.length == 0) {
                    revert("ERC721: transfer to non ERC721Receiver implementer");
                } else {
                    assembly {
                        revert(add(32, reason), mload(reason))
                    }
                }
            }
        } else {
            return true;
        }
    }

    /**
     * @dev Hook that is called before any token transfer. This includes minting
     * and burning.
     *
     * Calling conditions:
     *
     * - When `from` and `to` are both non-zero, ``from``'s `tokenId` will be
     * transferred to `to`.
     * - When `from` is zero, `tokenId` will be minted for `to`.
     * - When `to` is zero, ``from``'s `tokenId` will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {}
}
// File: ERC721Enumerable.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.6;

import "./ERC721.sol";
import "./IERC721Enumerable.sol";

/**
 * @dev This implements an optional extension of {ERC721} defined in the EIP that adds
 * enumerability of all the token ids in the contract as well as all token ids owned by each
 * account.
 */
abstract contract ERC721Enumerable is ERC721, IERC721Enumerable {
    // Mapping from owner to list of owned token IDs
    mapping(address => mapping(uint256 => uint256)) private _ownedTokens;

    // Mapping from token ID to index of the owner tokens list
    mapping(uint256 => uint256) private _ownedTokensIndex;

    // Array with all token ids, used for enumeration
    uint256[] private _allTokens;

    // Mapping from token id to position in the allTokens array
    mapping(uint256 => uint256) private _allTokensIndex;

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override(IERC165, ERC721) returns (bool) {
        return interfaceId == type(IERC721Enumerable).interfaceId
            || super.supportsInterface(interfaceId);
    }

    /**
     * @dev See {IERC721Enumerable-tokenOfOwnerByIndex}.
     */
    function tokenOfOwnerByIndex(address owner, uint256 index) public view virtual override returns (uint256) {
        require(index < ERC721.balanceOf(owner), "ERC721Enumerable: owner index out of bounds");
        return _ownedTokens[owner][index];
    }

    /**
     * @dev See {IERC721Enumerable-totalSupply}.
     */
    function totalSupply() public view virtual override returns (uint256) {
        return _allTokens.length;
    }

    /**
     * @dev See {IERC721Enumerable-tokenByIndex}.
     */
    function tokenByIndex(uint256 index) public view virtual override returns (uint256) {
        require(index < ERC721Enumerable.totalSupply(), "ERC721Enumerable: global index out of bounds");
        return _allTokens[index];
    }

    /**
     * @dev Hook that is called before any token transfer. This includes minting
     * and burning.
     *
     * Calling conditions:
     *
     * - When `from` and `to` are both non-zero, ``from``'s `tokenId` will be
     * transferred to `to`.
     * - When `from` is zero, `tokenId` will be minted for `to`.
     * - When `to` is zero, ``from``'s `tokenId` will be burned.
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(address from, address to, uint256 tokenId) internal virtual override {
        super._beforeTokenTransfer(from, to, tokenId);

        if (from == address(0)) {
            _addTokenToAllTokensEnumeration(tokenId);
        } else if (from != to) {
            _removeTokenFromOwnerEnumeration(from, tokenId);
        }
        if (to == address(0)) {
            _removeTokenFromAllTokensEnumeration(tokenId);
        } else if (to != from) {
            _addTokenToOwnerEnumeration(to, tokenId);
        }
    }

    /**
     * @dev Private function to add a token to this extension's ownership-tracking data structures.
     * @param to address representing the new owner of the given token ID
     * @param tokenId uint256 ID of the token to be added to the tokens list of the given address
     */
    function _addTokenToOwnerEnumeration(address to, uint256 tokenId) private {
        uint256 length = ERC721.balanceOf(to);
        _ownedTokens[to][length] = tokenId;
        _ownedTokensIndex[tokenId] = length;
    }

    /**
     * @dev Private function to add a token to this extension's token tracking data structures.
     * @param tokenId uint256 ID of the token to be added to the tokens list
     */
    function _addTokenToAllTokensEnumeration(uint256 tokenId) private {
        _allTokensIndex[tokenId] = _allTokens.length;
        _allTokens.push(tokenId);
    }

    /**
     * @dev Private function to remove a token from this extension's ownership-tracking data structures. Note that
     * while the token is not assigned a new owner, the `_ownedTokensIndex` mapping is _not_ updated: this allows for
     * gas optimizations e.g. when performing a transfer operation (avoiding double writes).
     * This has O(1) time complexity, but alters the order of the _ownedTokens array.
     * @param from address representing the previous owner of the given token ID
     * @param tokenId uint256 ID of the token to be removed from the tokens list of the given address
     */
    function _removeTokenFromOwnerEnumeration(address from, uint256 tokenId) private {
        // To prevent a gap in from's tokens array, we store the last token in the index of the token to delete, and
        // then delete the last slot (swap and pop).

        uint256 lastTokenIndex = ERC721.balanceOf(from) - 1;
        uint256 tokenIndex = _ownedTokensIndex[tokenId];

        // When the token to delete is the last token, the swap operation is unnecessary
        if (tokenIndex != lastTokenIndex) {
            uint256 lastTokenId = _ownedTokens[from][lastTokenIndex];

            _ownedTokens[from][tokenIndex] = lastTokenId; // Move the last token to the slot of the to-delete token
            _ownedTokensIndex[lastTokenId] = tokenIndex; // Update the moved token's index
        }

        // This also deletes the contents at the last position of the array
        delete _ownedTokensIndex[tokenId];
        delete _ownedTokens[from][lastTokenIndex];
    }

    /**
     * @dev Private function to remove a token from this extension's token tracking data structures.
     * This has O(1) time complexity, but alters the order of the _allTokens array.
     * @param tokenId uint256 ID of the token to be removed from the tokens list
     */
    function _removeTokenFromAllTokensEnumeration(uint256 tokenId) private {
        // To prevent a gap in the tokens array, we store the last token in the index of the token to delete, and
        // then delete the last slot (swap and pop).

        uint256 lastTokenIndex = _allTokens.length - 1;
        uint256 tokenIndex = _allTokensIndex[tokenId];

        // When the token to delete is the last token, the swap operation is unnecessary. However, since this occurs so
        // rarely (when the last minted token is burnt) that we still do the swap here to avoid the gas cost of adding
        // an 'if' statement (like in _removeTokenFromOwnerEnumeration)
        uint256 lastTokenId = _allTokens[lastTokenIndex];

        _allTokens[tokenIndex] = lastTokenId; // Move the last token to the slot of the to-delete token
        _allTokensIndex[lastTokenId] = tokenIndex; // Update the moved token's index

        // This also deletes the contents at the last position of the array
        delete _allTokensIndex[tokenId];
        _allTokens.pop();
    }
}
// File: IERC165.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.6;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}
// File: IERC721.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.6;

import "./IERC165.sol";

/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721 is IERC165 {
    /**
     * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
     */
    event Transfer(address indexed from, address indexed to, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
     */
    event Approval(address indexed owner, address indexed approved, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables or disables (`approved`) `operator` to manage all of its assets.
     */
    event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

    /**
     * @dev Returns the number of tokens in ``owner``'s account.
     */
    function balanceOf(address owner) external view returns (uint256 balance);

    /**
     * @dev Returns the owner of the `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function ownerOf(uint256 tokenId) external view returns (address owner);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be have been allowed to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Transfers `tokenId` token from `from` to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {safeTransferFrom} whenever possible.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Gives permission to `to` to transfer `tokenId` token to another account.
     * The approval is cleared when the token is transferred.
     *
     * Only a single account can be approved at a time, so approving the zero address clears previous approvals.
     *
     * Requirements:
     *
     * - The caller must own the token or be an approved operator.
     * - `tokenId` must exist.
     *
     * Emits an {Approval} event.
     */
    function approve(address to, uint256 tokenId) external;

    /**
     * @dev Returns the account approved for `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function getApproved(uint256 tokenId) external view returns (address operator);

    /**
     * @dev Approve or remove `operator` as an operator for the caller.
     * Operators can call {transferFrom} or {safeTransferFrom} for any token owned by the caller.
     *
     * Requirements:
     *
     * - The `operator` cannot be the caller.
     *
     * Emits an {ApprovalForAll} event.
     */
    function setApprovalForAll(address operator, bool _approved) external;

    /**
     * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
     *
     * See {setApprovalForAll}
     */
    function isApprovedForAll(address owner, address operator) external view returns (bool);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes calldata data
    ) external;
}
// File: IERC721Enumerable.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.6;

import "./IERC721.sol";

/**
 * @title ERC-721 Non-Fungible Token Standard, optional enumeration extension
 * @dev See https://eips.ethereum.org/EIPS/eip-721
 */
interface IERC721Enumerable is IERC721 {

    /**
     * @dev Returns the total amount of tokens stored by the contract.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns a token ID owned by `owner` at a given `index` of its token list.
     * Use along with {balanceOf} to enumerate all of ``owner``'s tokens.
     */
    function tokenOfOwnerByIndex(address owner, uint256 index) external view returns (uint256 tokenId);

    /**
     * @dev Returns a token ID at a given `index` of all the tokens stored by the contract.
     * Use along with {totalSupply} to enumerate all tokens.
     */
    function tokenByIndex(uint256 index) external view returns (uint256);
}
// File: IERC721Metadata.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.6;

import "./IERC721.sol";

/**
 * @title ERC-721 Non-Fungible Token Standard, optional metadata extension
 * @dev See https://eips.ethereum.org/EIPS/eip-721
 */
interface IERC721Metadata is IERC721 {
    /**
     * @dev Returns the token collection name.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the token collection symbol.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the Uniform Resource Identifier (URI) for `tokenId` token.
     */
    function tokenURI(uint256 tokenId) external view returns (string memory);
}
// File: IERC721Receiver.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.6;

/**
 * @title ERC721 token receiver interface
 * @dev Interface for any contract that wants to support safeTransfers
 * from ERC721 asset contracts.
 */
interface IERC721Receiver {
    /**
     * @dev Whenever an {IERC721} `tokenId` token is transferred to this contract via {IERC721-safeTransferFrom}
     * by `operator` from `from`, this function is called.
     *
     * It must return its Solidity selector to confirm the token transfer.
     * If any other value is returned or the interface is not implemented by the recipient, the transfer will be reverted.
     *
     * The selector can be obtained in Solidity with `IERC721.onERC721Received.selector`.
     */
    function onERC721Received(
        address operator,
        address from,
        uint256 tokenId,
        bytes calldata data
    ) external returns (bytes4);
}
// File: Ownable.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.6;

import "./Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _setOwner(_msgSender());
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _setOwner(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _setOwner(newOwner);
    }

    function _setOwner(address newOwner) private {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}
// File: PUZZLE.sol



// SPDX-License-Identifier: MIT
//
//  ██████╗ ██╗   ██╗███████╗███████╗██╗     ███████╗
//  ██╔══██╗██║   ██║╚══███╔╝╚══███╔╝██║     ██╔════╝
//  ██████╔╝██║   ██║  ███╔╝   ███╔╝ ██║     █████╗  
//  ██╔═══╝ ██║   ██║ ███╔╝   ███╔╝  ██║     ██╔══╝  
//  ██║     ╚██████╔╝███████╗███████╗███████╗███████╗
//  ╚═╝      ╚═════╝ ╚══════╝╚══════╝╚══════╝╚══════╝
//
// ******************************************************
//
//  REWARD -
//
//      Puzzle is a game rewarding its players.
//      32 ETH are available to claim, all written on-chain.
//      The bounty is split in 3 Chests, X (20 ETH), Y (10 ETH), Z (2 ETH).
//      It will be possible to open a chest only one time and this for ever. Once it is open,
//      the reward of the relative chest is gone.
//
//      TO BE ABLE TO CLAIM A REWARD YOU NEED TO GRIND FROM NO RANK TO SOLVER AND THEN SOLVE THE
//      ULTIMATE PUZZLE.
//
//      IT IS POSSIBLE TO CLAIM THE REWARD ONLY IF PUZZLE SOLD OUT, THE REWARD COME FROM THE SALE!
//
//      REWARD ARE CLAIMABLE ONLY ONCE, ONLY 3 WINNERS AT THE END.
//
//      Players do whatever they want of the reward, if you team up, feel free to split it.
//
// ******************************************************
//
//  GAME RULES -
//
//      RULES #0: Anyone can play Puzzle, the game is the contract.
//      
//      RULES #1: Only PUZZLE holders can verify steps on-chain and claim one of the reward at the end.
//
//      RULES #2: It is up to you to find out the purpose of Puzzle.
//
//      RULES #3: Puzzle have multiple steps waiting for you to complete. To pass from a step
//      to another, you have to verify data and then the contract will tell you if you are
//      good to go to the next step or have to retry.
//
//      RULES #4: To verify your data and complete a step, you need to use some
//      specific game functions. Those functions requiere two parameters.
//      
//      The parameters are: 
//          - uint tokenId, which represent an uint of a tokenId you own,
//          if you do not own the tokenId you request with, the function will throw into error.
//
//          - string data, which represent the data that have to generate the exact same hash of the
//          step you are on. The data to hash is a inline string without any separator.
//
//      The functions are:
//          - startGame(tokenId, data), this is the first step of the game, and this will
//          mark your NFT as a Player as soon as you complete this step.
//
//          - solveGame(tokenId, data), this is the second step of the game, and this will
//          mark your NFT as a Solver as soon as you complete this step.
//
//          - claimX(tokenId, data), claimY(tokenId, data), claimZ(tokenId, data), are all
//          functions of the ultimate step, callable only by people who already verified the two
//          previous steps. Those functions will instantly mark your NFT as a Keeper with
//          the identifier (X,Y,Z) of the reward you claimed and send you the ETH.
//
//      RULES #5: Technically it is ALMOST impossible to find out the data needed for the claimX/Y/Z
//      functions. As soon as you got the Solver grade, please, try your best to reach out
//      the author of this contract and get the final hint you need to claim your reward.
//
//      RULES #6: If you don't understand the game, look for friends and people
//      that would like to collaborate with you. Cooperation can be a huge advantage in the early
//      steps of the game. Keep in mind that as soon as you climb to Solver, it become a competition.
//
//      RULES #7: Have fun playing Puzzle! :)
//
// ******************************************************
//
//  ABOUT PUZZLE -
//
//      Puzzle is an experiment, a collection of 1024 nfts fully managed on-chain.
//      No official community behind, no website, no twitter, no discord.
//
//      You are free to create one and if you did you can try to contact the author of this contract
//      to allow him to join the party.
//
//      PUZZLE can be rare, there is a concept of rarity, but it do not have anything to do with the
//      game and owning a super rare PUZZLE doesn't change anything to the game nor to the wins.
//
//      The following data show the order of mint and the quantity for each PUZZLE:
//
//          _000000 = 535
//          _ffffff = 149
//          _9600ff = 116
//          _ff0000 = 115
//          _008aff = 58
//          _dbfffb = 45
//          _57c7f0 = 2
//          _a181c0 = 1
//          _56c6f0 = 1
//          _55c6f0 = 1
//          _a171c0 = 1
//
//      As you see, everything ends in a run.
//      But keep in mind, to play the game you need a PUZZLE and to win the game you need a PUZZLE.
//      Rarity doesn't matter to do both, but your PUZZLE nft will evolve while you go deeper in
//      the game steps. The evolution process will be write on chain, for ever, immutable. And the
//      last possible evolution given by claimX/Y/Z() functions are callable only once, for ever.
//
//      If you want to tryhard PUZZLE, a win on a super rare sound interesting.
//
//      If you have technical issue using this contract, please contact the author.
//
// ******************************************************
//
//  HOW TO MINT -
//      
//      Mint price is 0.1ETH/PUZZLE, you can mint up to 5 PUZZLEs in a batch (0.5 ETH).
//      To get the price value of your mint, use the function getPrice(uint quantity), price is given in wei.
//      Then to mint, use the function buyPUZZLE(uint quantity) with the value given by the getPrice() function.
//
//      FOR DUB HOLDERS:
//      If you own a DUB before the deployment of this contract, you are in the AllowList to claim 1 free PUZZLE.
//      Keep in mind that PUZZLEs are not reserved, if you do not claim your free PUZZLE before sold out,
//      you ain't be able to claim it later. To claim your PUZZLE use the claimPUZZLE() function.
//
//      175 ADRESSES HAVE BEEN ALLOWED, THE SNAPSHOT HAVE ALREADY HAPPEN, IF YOU GRAB A DUB NOW,
//      YOU AIN'T BE ABLE TO CLAIM A FREE PUZZLE.
//
// ******************************************************
//
//  HOW TO CUSTOMIZE -
//
//      As soon as you pass a step of the game by using the functions made for, you will be able to
//      update the MetaURI of your PUZZLE nft. The concept behind this is simple, there is 4 functions available,
//      all of those take the same tokenId parameter needed for the game functions. It have to be a tokenId you own.
//
//      Customizable functions are free but coast gas.
//      
//      The functions are:
//          - setBaseMetaURI(uint tokenId) which will set the base Image of your NFT, the one you got on mint
//
//          - setPlayerMetaURI(uint tokenId) which will set the base Image to Player Image, if you passed startGame() function
//
//          - setSolverMetaURI(uint tokenId) which will set the base Image to Solver Image, if you passed solveGame() function
//
//          - setKeeperMetaURI(uint tokenId) which will set the base Image to Keeper Image, if you passed claimX/Y/Z() function
//
// ******************************************************
//
//  HINTS -
//
//      You should watch for something that looks familliar.
//
//      "Consciousness isn't a journey upward, but a journey inward.
//      Not a pyramid, but a maze. Every choice could bring you closer to the center
//      or send you spiraling to the edges, to madness."
//
//      Follow @edoclip and use #WTFISPUZZLE on Twitter, sometimes hints drop from the nothing.
//
// ******************************************************
//
//  AUTHOR -
//
//      Ethereum: 0x064e92486a39ca396d7827d5182e1bb78a5a0d76
//      Twitter: twitter.com/edoclip
//      Discord: Qwerty#3641
//

pragma solidity ^0.8.6;

import './ERC721.sol';
import './ERC721Enumerable.sol';
import './Ownable.sol';
import './Base64.sol';
import './Strings.sol';

contract PUZZLE is ERC721, ERC721Enumerable, Ownable {
    
    string NAME = "Puzzle";
    string SYMBOL = "PUZZLE";
    string DESCRIPTION = "Welcome, this is Puzzle, a fully on-chain puzzle game experience.";
    
    // Rewards hold by chest (X/Y/Z) for a total of 32 ETH
    uint X = 20 ether;
    uint Y = 10 ether;
    uint Z = 2 ether;
    
    // First step to solve
    bytes32 public _metaMapHash = 0x6991a20ca075f4c7db3b84f65ba12ef1e17c36a59c36c0aa087a3fb3b7aba49d;
    // Second step to solve
    bytes32 public _metaKeyHash = 0x2fb145f188f7eb91450261f86badbc09b4f4de6fea934491a718e2814f94d50a;
    
    // Third step to solve and to claim rewards, contact author for hint
    bytes32 public _XHash = 0xcec6585c65952fb03c4f9bbf16256dc93014a85fba760a041f28810f45ce592a;
    bytes32 public _YHash = 0x875d8f442a250f03587c996399e4ac0aab43caad2c6fc23ea72c4dd69a84c607;
    bytes32 public _ZHash = 0x7cb028a240de901d2e2ab9dd335c8e0a621f793bcfaad9217d3005a817fa6923;
    
    uint public _000000 = 535;
    uint public _ffffff = 149;
    uint public _9600ff = 116;
    uint public _ff0000 = 115;
    uint public _008aff = 58;
    uint public _dbfffb = 45;
    uint public _57c7f0 = 2;
    uint public _a181c0 = 1;
    uint public _56c6f0 = 1;
    uint public _55c6f0 = 1;
    uint public _a171c0 = 1;
    
    // map tokenId to metaCore
    mapping (uint => string) public _metaCore;
    
    // map tokenId to string for metaInfo
    mapping (uint => string) _metaName;
    mapping (uint => string) _metaAttributes;
    mapping (uint => string) _metaURI;
    
    // map tokenId to bool for metaRank
    mapping (uint => bool) public _metaPlayers;
    mapping (uint => bool) public _metaSolvers;
    mapping (uint => bool) public _metaKeepers;
    
    // map tokenId to chest identifier (X/Y/Z)
    mapping (uint => string) _keeperOf;
    
    // map chest hash (XHash,YHash,ZHash) to address of the claimer
    mapping (bytes32 => address) public _XClaimed;
    mapping (bytes32 => address) public _YClaimed;
    mapping (bytes32 => address) public _ZClaimed;
    
    uint public _salePrice = 0.1 ether;
    bool public _saleState = false;
    uint public _maxSupply = 1024;
    
    bool allAssigned = false;
    uint nextIndexToAssign = 0;
    
    string JSON = 'data:application/json;base64,';
    string SVG = 'data:image/svg+xml;base64,';
    
    mapping (address => bool) public _allowList;
        
    event PUZZLEMinted(uint tokenId, address owner);
    
    event PLAYERLoaded(uint tokenId, address owner);
    event SOLVERLoaded(uint tokenId, address owner);
    event KEEPERLoaded(uint tokenId, address owner, string claimed);
    
    event XClaimed(uint tokenId, address owner);
    event YClaimed(uint tokenId, address owner);
    event ZClaimed(uint tokenId, address owner);
    
    event MetaURIUpdated(uint tokenId, address owner);
    
    modifier ownThePUZZLE(uint _tokenId, address _from) {
        _;
        require(_exists(_tokenId), "error tokenId");
        require(ERC721.ownerOf(_tokenId) == _from, "error ownerOf");
    }

    constructor() ERC721(NAME, SYMBOL) {}

    // Sale functions  ****************************************************
    
    function setSaleState()
    public onlyOwner {
        _saleState = _saleState ? false : true;
    }
    
    function getPUZZLEPrice(uint quantity)
    public view returns (uint price) {
        price = quantity * _salePrice;
    }
    
    function withdrawEquity()
    public onlyOwner {
        uint balance = address(this).balance;
        require(payable(msg.sender).send(balance));
    }
    
    // Mint functions  ***************************************************
    
    function _mintPUZZLE(uint tokenId, address to)
    private {
        _setMetaName(tokenId);
        _setMetaOrigins(tokenId);
        _setBaseMetaURI(tokenId);
        _safeMint(to, tokenId);
        emit PUZZLEMinted(tokenId, to);
    }
    
    function _setPUZZLE(address to)
    private {
        require(nextIndexToAssign != _maxSupply, "error _maxSupply");
        require(!allAssigned, "error allAssigned");
        
        uint tokenId = nextIndexToAssign;
        nextIndexToAssign++;
        
        if (nextIndexToAssign == _maxSupply) {
            allAssigned = true;
            _saleState = false;
        }
        
        if (_000000 != 0) {
            _000000--;
            _metaCore[tokenId] = "000000";
            _mintPUZZLE(tokenId, to);
            return;
        }
        
        if (_ffffff != 0) {
            _ffffff--;
            _metaCore[tokenId] = "ffffff";
            _mintPUZZLE(tokenId, to);
            return;
        }
        
        if (_9600ff != 0) {
            _9600ff--;
            _metaCore[tokenId] = "9600ff";
            _mintPUZZLE(tokenId, to);
            return;
        }
        
        if (_ff0000 != 0) {
            _ff0000--;
            _metaCore[tokenId] = "ff0000";
            _mintPUZZLE(tokenId, to);
            return;
        }
        
        if (_008aff != 0) {
            _008aff--;
            _metaCore[tokenId] = "008aff";
            _mintPUZZLE(tokenId, to);
            return;
        }
        
        if (_dbfffb != 0) {
            _dbfffb--;
            _metaCore[tokenId] = "dbfffb";
            _mintPUZZLE(tokenId, to);
            return;
        }
        
        if (_57c7f0 != 0) {
            _57c7f0--;
            _metaCore[tokenId] = "57c7f0";
            _mintPUZZLE(tokenId, to);
            return;
        }
        
        if (_a181c0 != 0) {
            _a181c0--;
            _metaCore[tokenId] = "a181c0";
            _mintPUZZLE(tokenId, to);
            return;
        }
        
        if (_56c6f0 != 0) {
            _56c6f0--;
            _metaCore[tokenId] = "56c6f0";
            _mintPUZZLE(tokenId, to);
            return;
        }
        
        if (_55c6f0 != 0) {
            _55c6f0--;
            _metaCore[tokenId] = "55c6f0";
            _mintPUZZLE(tokenId, to);
            return;
        }
        
        if (_a171c0 != 0) {
            _a171c0--;
            _metaCore[tokenId] = "a171c0";
            _mintPUZZLE(tokenId, to);
            return;
        }
    }

    function buyPUZZLE(uint quantity)
    public payable {
        require(_saleState, "error _saleState");
        require(quantity > 0 && quantity <= 5, "error quantity");
        require(msg.value == getPUZZLEPrice(quantity), "error _salePrice");
        
        uint i = 0;
        while(quantity > i++) _setPUZZLE(msg.sender);
    }
    
    function claimPUZZLE()
    public {
        require(_saleState, "error _saleState");
        require(_allowList[msg.sender], "error allowList");
        
        _allowList[msg.sender] = false;
        _setPUZZLE(msg.sender);
    }
    
    // Game functions  ****************************************************
    
    // To enter the game, find the data that match with _metaMapHash
    function startGame(uint tokenId, string memory data)
    public ownThePUZZLE(tokenId, msg.sender) {
        string memory tmpAttributes = string(abi.encodePacked(
            '{"trait_type":"Core","value":"_',_metaCore[tokenId],'"},',
            '{"trait_type":"Player","value":"True"},',
            '{"trait_type":"Solver","value":"False"},',
            '{"trait_type":"Keeper","value":"False"},',
            '{"trait_type":"X","value":"False"},',
            '{"trait_type":"Y","value":"False"},',
            '{"trait_type":"Z","value":"False"}'
        ));
        
        require(!_metaPlayers[tokenId], "error _metaPlayers");
        require(keccak256(abi.encodePacked(data)) == _metaMapHash, "error _metaMapHash");
        
        _metaPlayers[tokenId] = true;
        _metaAttributes[tokenId] = tmpAttributes;
        
        emit PLAYERLoaded(tokenId, msg.sender);
    }
    
    // To solve the game, find the data that match with _metaKeyHash
    function solveGame(uint tokenId, string memory data)
    public ownThePUZZLE(tokenId, msg.sender)  {
        string memory tmpAttributes = string(abi.encodePacked(
            '{"trait_type":"Core","value":"_',_metaCore[tokenId],'"},',
            '{"trait_type":"Player","value":"True"},',
            '{"trait_type":"Solver","value":"True"},',
            '{"trait_type":"Keeper","value":"False"},',
            '{"trait_type":"X","value":"False"},',
            '{"trait_type":"Y","value":"False"},',
            '{"trait_type":"Z","value":"False"}'
        ));
        
        require(_metaPlayers[tokenId], "error _metaPlayers");
        require(!_metaSolvers[tokenId], "error _metaSolvers");
        require(keccak256(abi.encodePacked(data)) == _metaKeyHash, "error _metaKeyHash");
        
        _metaSolvers[tokenId] = true;
        _metaAttributes[tokenId] = tmpAttributes;
        
        emit SOLVERLoaded(tokenId, msg.sender);
    }
    
    // To claim X, find the data that match with _XHash, contact author for hint
    function claimX(uint tokenId, string memory data)
    public ownThePUZZLE(tokenId, msg.sender)  {
        string memory tmpAttributes = string(abi.encodePacked(
            '{"trait_type":"Core","value":"_',_metaCore[tokenId],'"},',
            '{"trait_type":"Player","value":"True"},',
            '{"trait_type":"Solver","value":"True"},',
            '{"trait_type":"Keeper","value":"True"},',
            '{"trait_type":"X","value":"True"},',
            '{"trait_type":"Y","value":"False"},',
            '{"trait_type":"Z","value":"False"}'
        ));
        
        require(allAssigned, "error allAssigned");
        require(address(this).balance > X, "error balance - contact author");
        require(_XClaimed[_XHash] == address(0), "error _XClaimed");
        require(_metaPlayers[tokenId], "error _metaPlayers");
        require(_metaSolvers[tokenId], "error _metaSolvers");
        require(!_metaKeepers[tokenId], "error _metaKeepers");
        require(keccak256(abi.encodePacked(data)) == _XHash, "error _XHash");
        
        _metaKeepers[tokenId] = true;
        _XClaimed[_XHash] = msg.sender;
        _keeperOf[tokenId] = 'X';
        _metaAttributes[tokenId] = tmpAttributes;
        
        uint tmp = X;
        X = 0;
        require(payable(msg.sender).send(tmp));
        
        emit KEEPERLoaded(tokenId, msg.sender, 'X');
    }
    
    // To claim Y, find the data that match with _YHash, contact author for hint
    function claimY(uint tokenId, string memory data)
    public ownThePUZZLE(tokenId, msg.sender)  {
        string memory tmpAttributes = string(abi.encodePacked(
            '{"trait_type":"Core","value":"_',_metaCore[tokenId],'"},',
            '{"trait_type":"Player","value":"True"},',
            '{"trait_type":"Solver","value":"True"},',
            '{"trait_type":"Keeper","value":"True"},',
            '{"trait_type":"X","value":"False"},',
            '{"trait_type":"Y","value":"True"},',
            '{"trait_type":"Z","value":"False"}'
        ));
        
        require(allAssigned, "error allAssigned");
        require(address(this).balance > Y, "error balance - retry later");
        require(_YClaimed[_YHash] == address(0), "error _YClaimed");
        require(_metaPlayers[tokenId], "error _metaPlayers");
        require(_metaSolvers[tokenId], "error _metaSolvers");
        require(!_metaKeepers[tokenId], "error _metaKeepers");
        require(keccak256(abi.encodePacked(data)) == _YHash, "error _YHash");
        
        _metaKeepers[tokenId] = true;
        _YClaimed[_YHash] = msg.sender;
        _keeperOf[tokenId] = 'Y';
        _metaAttributes[tokenId] = tmpAttributes;
        
        uint tmp = Y;
        Y = 0;
        require(payable(msg.sender).send(tmp));
        
        emit KEEPERLoaded(tokenId, msg.sender, 'Y');
    }
    
    // To claim Z, find the data that match with _ZHash, contact author for hint
    function claimZ(uint tokenId, string memory data)
    public ownThePUZZLE(tokenId, msg.sender)  {
        string memory tmpAttributes = string(abi.encodePacked(
            '{"trait_type":"Core","value":"_',_metaCore[tokenId],'"},',
            '{"trait_type":"Player","value":"True"},',
            '{"trait_type":"Solver","value":"True"},',
            '{"trait_type":"Keeper","value":"True"},',
            '{"trait_type":"X","value":"False"},',
            '{"trait_type":"Y","value":"False"},',
            '{"trait_type":"Z","value":"True"}'
        ));
        
        require(allAssigned, "error allAssigned");
        require(address(this).balance > Z, "error balance - retry later");
        require(_ZClaimed[_ZHash] == address(0), "error _ZClaimed");
        require(_metaPlayers[tokenId], "error _metaPlayers");
        require(_metaSolvers[tokenId], "error _metaSolvers");
        require(!_metaKeepers[tokenId], "error _metaKeepers");
        require(keccak256(abi.encodePacked(data)) == _ZHash, "error _ZHash");
        
        _metaKeepers[tokenId] = true;
        _ZClaimed[_ZHash] = msg.sender;
        _keeperOf[tokenId] = 'Z';
        _metaAttributes[tokenId] = tmpAttributes;
        
        uint tmp = Z;
        Z = 0;
        require(payable(msg.sender).send(tmp));
        
        emit KEEPERLoaded(tokenId, msg.sender, 'Z');
    }
    
    // Meta functions  ****************************************************
    
    function _setMetaName(uint tokenId)
    private {
        string memory tmp;
        if (tokenId == 0) tmp = '0000';
        if (tokenId >= 1000) tmp = Strings.toString(tokenId);
        if (tokenId < 1000) tmp = string(abi.encodePacked('0', Strings.toString(tokenId)));
        if (tokenId < 100) tmp = string(abi.encodePacked('00', Strings.toString(tokenId)));
        if (tokenId < 10) tmp = string(abi.encodePacked('000', Strings.toString(tokenId)));
        
        _metaName[tokenId] = string(abi.encodePacked(
            NAME, ' #', tmp
        ));
    }
    
    function _setMetaOrigins(uint tokenId)
    private {
        _metaAttributes[tokenId] = string(abi.encodePacked(
            '{"trait_type":"Core","value":"_',_metaCore[tokenId],'"},',
            '{"trait_type":"Player","value":"False"},',
            '{"trait_type":"Solver","value":"False"},',
            '{"trait_type":"Keeper","value":"False"},',
            '{"trait_type":"X","value":"False"},',
            '{"trait_type":"Y","value":"False"},',
            '{"trait_type":"Z","value":"False"}'
        ));
    }
    
    function _setBaseMetaURI(uint tokenId)
    private {
        _metaURI[tokenId] = string(abi.encodePacked(
            SVG, 
            Base64.encode(bytes(string(abi.encodePacked(
                '<svg xmlns="http://www.w3.org/2000/svg" preserveAspectRatio="xMinYMin meet" viewBox="0 0 350 350">',
                '<rect width="100%" height="100%" fill="#',_metaCore[tokenId],'"/>',
                '</svg>'
            ))))
        ));
    }
    
    function setBaseMetaURI(uint tokenId)
    public ownThePUZZLE(tokenId, msg.sender) {
        _setBaseMetaURI(tokenId);
        
        emit MetaURIUpdated(tokenId, msg.sender);
    }
    
    function setPlayerMetaURI(uint tokenId)
    public ownThePUZZLE(tokenId, msg.sender) {
        require(_metaPlayers[tokenId], "error _metaPlayers");
        
        _metaURI[tokenId] = string(abi.encodePacked(
            SVG, 
            Base64.encode(bytes(string(abi.encodePacked(
                '<svg xmlns="http://www.w3.org/2000/svg" preserveAspectRatio="xMinYMin meet" viewBox="0 0 350 350">',
                '<style xmlns="http://www.w3.org/2000/svg">.base { fill: #ffffff; font-family: serif; font-size: 16px; }</style>',
                '<rect width="100%" height="100%" fill="#000000"/>',
                '<text x="10" y="20" class="base">_', _metaCore[tokenId],'</text>',
                '<text x="10" y="40" class="base">Player</text>',
                '</svg>'
            ))))
        ));
        
        emit MetaURIUpdated(tokenId, msg.sender);
    }
    
    function setSolverMetaURI(uint tokenId)
    public ownThePUZZLE(tokenId, msg.sender) {
        require(_metaPlayers[tokenId], "error _metaPlayers");
        require(_metaSolvers[tokenId], "error _metaSolvers");
        
        _metaURI[tokenId] = string(abi.encodePacked(
            SVG, 
            Base64.encode(bytes(string(abi.encodePacked(
                '<svg xmlns="http://www.w3.org/2000/svg" preserveAspectRatio="xMinYMin meet" viewBox="0 0 350 350">',
                '<style xmlns="http://www.w3.org/2000/svg">.base { fill: #ffffff; font-family: serif; font-size: 16px; }</style>',
                '<rect width="100%" height="100%" fill="#000000"/>',
                '<text x="10" y="20" class="base">_', _metaCore[tokenId],'</text>',
                '<text x="10" y="40" class="base">Solver</text>',
                '</svg>'
            ))))
        ));
        
        emit MetaURIUpdated(tokenId, msg.sender);
    }
    
    function setKeeperMetaURI(uint tokenId)
    public ownThePUZZLE(tokenId, msg.sender) {
        require(_metaPlayers[tokenId], "error _metaPlayers");
        require(_metaSolvers[tokenId], "error _metaSolvers");
        require(_metaKeepers[tokenId], "error _metaKeepers");
        
        _metaURI[tokenId] = string(abi.encodePacked(
            SVG, 
            Base64.encode(bytes(string(abi.encodePacked(
                '<svg xmlns="http://www.w3.org/2000/svg" preserveAspectRatio="xMinYMin meet" viewBox="0 0 350 350">',
                '<style xmlns="http://www.w3.org/2000/svg">.base { fill: #000000; font-family: serif; font-size: 16px; }</style>',
                '<rect width="100%" height="100%" fill="#ffffff"/>',
                '<text x="10" y="20" class="base">_', _metaCore[tokenId],'</text>',
                '<text x="10" y="40" class="base">Keeper</text>',
                '<text x="10" y="60" class="base">', _keeperOf[tokenId],'</text>',
                '</svg>'
            ))))
        ));
        
        emit MetaURIUpdated(tokenId, msg.sender);
    }
    
    // Allowance functions  ***********************************************
    
    function setBatchAllowance(address[] memory batch)
    public onlyOwner {
        uint len = batch.length;
        require(len > 0, "error len");
        
        uint i = 0;
        while (i < len) {
            _allowList[batch[i]] = _allowList[batch[i]] ?
            false : true;
            i++;
        }
    }

    // Token URI functions ************************************************
    
    function _tokenData(uint256 tokenId)
    private view returns (string memory result) {
        result = string(abi.encodePacked(
            '{',
            '"name":"',_metaName[tokenId],'",',
            '"description":"',DESCRIPTION,'",',
            '"attributes":[',_metaAttributes[tokenId],'],',
            '"image":"',_metaURI[tokenId],'"',
            '}'
        ));
    }

    function tokenURI(uint256 tokenId)
    public view virtual override returns (string memory result) {
        require(_exists(tokenId), "ERC721Metadata: URI query for nonexistent token");
        
        result = string(abi.encodePacked(
            JSON,
            Base64.encode(bytes(string(abi.encodePacked(
                _tokenData(tokenId)
            ))))
        ));
    }
    
    // Contract URI functions *********************************************
    
    function _contractData()
    private view returns (string memory result) {
        result = string(abi.encodePacked(
            '{',
            '"name":"',NAME,'",',
            '"description":"',DESCRIPTION,'",',
            '"image":"ipfs://bafybeicbfc77qsny2e7ctv76nmctbitlun6uuuhhisxrqe5fy6qi2wqxlu/",',
            '"seller_fee_basis_points":"500",',
            '"fee_recipient":"0x064e92486a39ca396d7827d5182e1bb78a5a0d76",',
            '"image":"ipfs://bafybeicbfc77qsny2e7ctv76nmctbitlun6uuuhhisxrqe5fy6qi2wqxlu/"',
            '}'
        ));
    }
    
    function contractURI()
    public view virtual returns (string memory result) {
        result = string(abi.encodePacked(
            JSON,
            Base64.encode(bytes(string(abi.encodePacked(
                _contractData()
            ))))
        ));
    }

    // ERC721 Spec functions **********************************************
    
    function _beforeTokenTransfer(address from, address to, uint tokenId)
    internal override(ERC721, ERC721Enumerable) {
        super._beforeTokenTransfer(from, to, tokenId);
    }

    function supportsInterface(bytes4 interfaceId)
    public view override(ERC721, ERC721Enumerable)
    returns (bool) {
        return super.supportsInterface(interfaceId);
    }
}
// File: SafeMath.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.6;

// CAUTION
// This version of SafeMath should only be used with Solidity 0.8 or later,
// because it relies on the compiler's built in overflow checks.

/**
 * @dev Wrappers over Solidity's arithmetic operations.
 *
 * NOTE: `SafeMath` is no longer needed starting with Solidity 0.8. The compiler
 * now has built in overflow checking.
 */
library SafeMath {
    /**
     * @dev Returns the addition of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function tryAdd(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            uint256 c = a + b;
            if (c < a) return (false, 0);
            return (true, c);
        }
    }

    /**
     * @dev Returns the substraction of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function trySub(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            if (b > a) return (false, 0);
            return (true, a - b);
        }
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function tryMul(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            // Gas optimization: this is cheaper than requiring 'a' not being zero, but the
            // benefit is lost if 'b' is also tested.
            // See: https://github.com/OpenZeppelin/openzeppelin-contracts/pull/522
            if (a == 0) return (true, 0);
            uint256 c = a * b;
            if (c / a != b) return (false, 0);
            return (true, c);
        }
    }

    /**
     * @dev Returns the division of two unsigned integers, with a division by zero flag.
     *
     * _Available since v3.4._
     */
    function tryDiv(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            if (b == 0) return (false, 0);
            return (true, a / b);
        }
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers, with a division by zero flag.
     *
     * _Available since v3.4._
     */
    function tryMod(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            if (b == 0) return (false, 0);
            return (true, a % b);
        }
    }

    /**
     * @dev Returns the addition of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `+` operator.
     *
     * Requirements:
     *
     * - Addition cannot overflow.
     */
    function add(uint256 a, uint256 b) internal pure returns (uint256) {
        return a + b;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting on
     * overflow (when the result is negative).
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b) internal pure returns (uint256) {
        return a - b;
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `*` operator.
     *
     * Requirements:
     *
     * - Multiplication cannot overflow.
     */
    function mul(uint256 a, uint256 b) internal pure returns (uint256) {
        return a * b;
    }

    /**
     * @dev Returns the integer division of two unsigned integers, reverting on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator.
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b) internal pure returns (uint256) {
        return a / b;
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * reverting when dividing by zero.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b) internal pure returns (uint256) {
        return a % b;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting with custom message on
     * overflow (when the result is negative).
     *
     * CAUTION: This function is deprecated because it requires allocating memory for the error
     * message unnecessarily. For custom revert reasons use {trySub}.
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(
        uint256 a,
        uint256 b,
        string memory errorMessage
    ) internal pure returns (uint256) {
        unchecked {
            require(b <= a, errorMessage);
            return a - b;
        }
    }

    /**
     * @dev Returns the integer division of two unsigned integers, reverting with custom message on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(
        uint256 a,
        uint256 b,
        string memory errorMessage
    ) internal pure returns (uint256) {
        unchecked {
            require(b > 0, errorMessage);
            return a / b;
        }
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * reverting with custom message when dividing by zero.
     *
     * CAUTION: This function is deprecated because it requires allocating memory for the error
     * message unnecessarily. For custom revert reasons use {tryMod}.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(
        uint256 a,
        uint256 b,
        string memory errorMessage
    ) internal pure returns (uint256) {
        unchecked {
            require(b > 0, errorMessage);
            return a % b;
        }
    }
}
// File: Strings.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.6;

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _HEX_SYMBOLS = "0123456789abcdef";

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        // Inspired by OraclizeAPI's implementation - MIT licence
        // https://github.com/oraclize/ethereum-api/blob/b42146b063c7d6ee1358846c198246239e9360e8/oraclizeAPI_0.4.25.sol

        if (value == 0) {
            return "0";
        }
        uint256 temp = value;
        uint256 digits;
        while (temp != 0) {
            digits++;
            temp /= 10;
        }
        bytes memory buffer = new bytes(digits);
        while (value != 0) {
            digits -= 1;
            buffer[digits] = bytes1(uint8(48 + uint256(value % 10)));
            value /= 10;
        }
        return string(buffer);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        if (value == 0) {
            return "0x00";
        }
        uint256 temp = value;
        uint256 length = 0;
        while (temp != 0) {
            length++;
            temp >>= 8;
        }
        return toHexString(value, length);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _HEX_SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }
}